บทที่ 9
การอินทีเกรท

การอินทีเกรทที่ง่ายที่สุดก็คือ การอินทีเกรทโดยการแทนฟังก์ชันด้วยโพลีโนเมียล

$$I = \int_a^b f(x)\,dx \cong \int_a^b f_n(x)\,dx$$

เมื่อ

$$f(x) = a_0 + a_1 x + ... + a_{n-1} x^{n-1} + a_n x^n$$



รูปที่ 9.1 การประมาณอินทีเกรทด้วยพื้นที่ใต้กราฟ a) เส้นตรง b) เส้นพาราโบลา



รูปที่ 9.2 การประมาณอินทีเกรทด้วยพื้นที่ใต้กราฟเส้นตรงสามเส้น

ตัวอย่างที่ 9.1 จงอินทีเกรทสมการต่อไปนี้

a) $\int_0^3 \left(1-e^{-x}\right)dx$

b) $\int_{-2}^4 \left(1-x-4x^3+x^5\right)dx$

c) $\int_0^{\pi/2} \left(8+4\sin x\right)dx$

วิธีทำ

a) $\int_0^3 \left(1-e^{-x}\right)dx$ = 2.05

b) $\int_{-2}^4 \left(1-x-4x^3+x^5\right)dx$ = 432

c) $\int_0^{\pi/2} \left(8+4\sin x\right)dx$ = 16.566

9.1 ทราเปซอยดอลรูล (Trapezoidal rule)

อาศัยโพลีโนเมียลลำดับแรก

$$I = \int_a^b f(x)\,dx \cong \int_a^b f_1(x)\,dx$$

เมื่อ

$$f_1(x) = f(a) + \frac{f(b)-f(a)}{b-a}(x-a)$$

$$I = \int_a^b \left[f(a) + \frac{f(b)-f(a)}{b-a}(x-a) \right] dx$$

จะได้

$$I = (b-a)\frac{f(a)+f(b)}{2}$$

เรียกว่า Trapezoidal rule



รูปที่ 9.3 แสดง Trapezoidal rule

โดยมีความคลาดเคลื่อน

$$E_r = -\frac{1}{12} f^3(\xi)(b-a)^3$$

เมื่อ ξ อยู่ระหว่าง a และ b แต่ถ้าต้องการให้ได้ความถูกต้องแม่นยำมากยิ่งขึ้น ให้แบ่งช่วงของการอินทีเกรทเป็นส่วนเล็กหลาย ๆ ส่วน จากนั้นให้ใช้ทราเปซอยดอลรูลกับแต่ละส่วนเรียกว่า Multiple – Trapezoidal rule

ตัวอย่างที่ 9.2 จงอาศัย Trapezoidal rule. อินทิเกรตสมการ

$$f_1(x) = 0.2 + 25x - 200x^2 + 675x^3 - 900x^4 + 400x^5$$

จาก a = 0 ถึง b = 0.8 และจงเขียนโปรแกรมโดยใช้วิธีทราเปซอยดอล และวิธี Multiple – Trapezoidal

วิธีทำ

$$f(0) = 0.2$$

$$f\ (0.8) = 0.232$$

$$I \cong (0.8)\frac{0.2 + 0.232}{2} = 0.1728$$

ค่าที่แท้จริงคือ 1.640533

$E_r = 1.640533 - 0.1728 = 1.467733$ คิดเป็นerror 89.5%

โปรแกรมที่ใช้ในการคำนวณ คือ

```
      PROGRAM TRAPZ
      REAL A,B,OLDVAL,NEWVAL,H,INTEGR
      INTEGER N
      PRINT *,'ENTER LEFT AND RIGHT ENDPOINTS OF INTERVAL'
      PRINT *,'OF INTEGRATION. ALSO ENTER ACCURACY DESIRED.'
      PRINT *,'ENTER A,B,N '
      READ *,A,B,N
      H=(B-A)/N
      OLDVAL = (B-A)*(F(A)+F(B))/2.
      SUM = (F(A)+F(B))/2
      DO 10 I=1,N-1
        SUM = SUM+F(A+I*H)
10     CONTINUE
      INTEGR =H*SUM
      NEWVAL = INTEGR
```

```
      WRITE(*,100)N-1,OLDVAL
      WRITE(*,200)NEWVAL
100   FORMAT(/,7X,'THE INTEGRAL VALUE AFTER ',I3,'  ITERATION',//,
    +      7X,'BY THE TRAPEZOIDAL RULE IS :',F10.6,///)
200   FORMAT( 7X,'THE VALUE OF THE INTEGRAL BY',//,
    +     7X,'THE MULTIPLE-TRAPEZOIDAL RULE IS :',F10.6)
      STOP
      END
*------------------------------------------------------------------
      FUNCTION F(X)
      REAL X
      F = 0.2+25.*X-200.*X**2.+675.*X**3.-900.*X**4.+400.*X**5.
      END
*------------------------------------------------------------------
```

เอาท์พุทจากการคำนวณ คือ



```
 ENTER LEFT AND RIGHT ENDPOINTS OF INTERVAL
 OF INTEGRATION. ALSO ENTER ACCURACY DESIRED.
 ENTER A,B,N
0
0.8
10

       THE INTEGRAL VALUE AFTER   9  ITERATION

       BY THE TRAPEZOIDAL RULE IS :  0.172800


       THE VALUE OF THE INTEGRAL BY

       THE MULTIPLE-TRAPEZOIDAL RULE IS :  1.615042
```

ซึ่งจะได้ค่าอินทีเกรท คือ 0.1728 ตรงกับการคำนวณด้วยมือ

แต่ถ้าใช้ multiple trapezoidal rule จะได้ค่าอินทีเกรทคือ 1.615042 ซึ่งมีค่าใกล้เคียงกับค่าที่แท้จริงมาก

ตัวอย่างที่ 9.3 อาศัยวิธีทราเปซอยดอล และ multiple trapezoidal คำนวณอินทีเกรทในตัวอย่าง 9.1

วิธีทำ ใช้โปรแกรมคำนวณอินทีเกรทโดยวิธีทราเปซอยดอลในตัวอย่าง 9.1 จะได้เอาท์พุทคือ

a)

```
 ENTER LEFT AND RIGHT ENDPOINTS OF INTERVAL
 OF INTEGRATION. ALSO ENTER ACCURACY DESIRED.
 ENTER A,B,N
0
3
10

       THE INTEGRAL VALUE AFTER       9  ITERATION

       BY THE TRAPEZOIDAL RULE IS :  1.425319


       THE VALUE OF THE INTEGRAL BY

       THE MULTIPLE-TRAPEZOIDAL RULE IS :  2.042671
```

b)

```
 ENTER LEFT AND RIGHT ENDPOINTS OF INTERVAL
 OF INTEGRATION. ALSO ENTER ACCURACY DESIRED.
 ENTER A,B,N
-2
4
10

       THE INTEGRAL VALUE AFTER   9  ITERATION

       BY THE TRAPEZOIDAL RULE IS :     2304.000000


       THE VALUE OF THE INTEGRAL BY

       THE MULTIPLE-TRAPEZOIDAL RULE IS :      463.550476
```

c)

```
 ENTER LEFT AND RIGHT ENDPOINTS OF INTERVAL
 OF INTEGRATION. ALSO ENTER ACCURACY DESIRED.
 ENTER A,N
0
1000

       THE INTEGRAL VALUE AFTER 999  ITERATION

       BY THE TRAPEZOIDAL RULE IS : 15.707964

       THE VALUE OF THE INTEGRAL BY

       THE MULTIPLE-TRAPEZOIDAL RULE IS : 16.566374
```

จะเห็นได้ว่า ถ้าแบ่ง segment จำนวนน้อยค่าที่ได้จะผิดพลาดไปมาก ในข้อ c) ใช้จำนวน segment = 1,000 จึงมีค่าใกล้เคียงกับค่าที่แท้จริง อย่างไรก็ตาม ถ้า segment มีจำนวนมาก การคำนวณด้วยเครื่องคิดเลขจะยุ่งยากและใช้เวลานานมาก

9.2 วิธีซิมป์สัน (Simpson's Rule)

อาศัยโพลีโนเมียลลำดับสูง มี 2 วิธี คือ

วิธีที่ 1 หลักการซิมป์สัน 1/3



รูปที่ 9.3 หลักการซิมป์สัน 1/3

สูตรที่ใช้ในการคำนวณคือ

$$I = (b-a)\frac{f(x_0)+4f(x_1)+f(x_2)}{6}$$

ความคลาดเคลื่อน

$$E_r = -\frac{(b-a)^5}{2880} f^4(\xi)$$

สำหรับการใช้หลักการซิมป์สัน 1/3 หลาย ๆ ครั้ง ครั้งละหนึ่งส่วน



รูปที่ 9.4 การใช้หลักการซิมป์สัน 1/3 แต่ละส่วน

$$h = -\frac{(b-a)}{n}$$

$$I \cong 2h\frac{f(x_0)+4f(x_1)+f(x_2)}{6} + 2h\frac{f(x_2)+4f(x_3)+f(x_4)}{6} + \ldots + 2h\frac{f(x_{n-2})+4f(x_{n-1})+f(x_n)}{6}$$

หรือ

$$I \cong (b-a)\frac{f(x_0)+4\sum_{i=1,3,5}^{n-1} f(x_i)+2\sum_{j=2,4,6}^{n-2} f(x_j)\,f(x_n)}{3n}$$

โดยมีความคลาดเคลื่อน

$$E_a = -\frac{(b-a)^5}{180n}\overline{f}^4$$

วิธีที่ 2 หลักการซิมป์สัน 3/8

สมการที่ใช้คำนวณคือ

$$I \cong (b-a)\frac{f(x_0)+3f(x_1)+3f(x_2)+f(x_3)}{8}$$

โดยมีความคลาดเคลื่อน

$$E_a = -\frac{(b-a)^5}{6480}\bar{f}^4(\xi)$$

ตัวอย่างที่ 9.4 จงหาค่าอินทีเกรทของสมการ โดยวิธีซิมป์สัน 1/3

$$f(x) = 0.2 + 25x - 200x^2 + 675x^3 - 900x^4 + 400x^5$$

จาก a = 0 ถึง b = 0.8 คำตอบที่ถูกต้องคือ 1.640533

วิธีทำ ใช้ n = 4 (h = 0.2)

f(0) = 0.2 f(0.2) = 1.288

f(0.4) = 2.456 f(0.6) = 3.464

f(0.8) = 0.232

จะได้

$$I = 0.8\frac{0.2+4(1.288+3.464)+2(2.456)+0.232}{12}$$

$$= 1.623467$$

มีค่าใกล้เคียงกับค่าที่แท้จริง

ความคลาดเคลื่อน

$$E_a = -\frac{(0.8)^2}{180(4)^4}(-2400)$$

$$= 0.017067$$

โปรแกรมที่ใช้ในการคำนวณคือ

```
      PROGRAM SIMPSN
      REAL A,B,INTEGL,SUMODD,SUMEVEN,H
      INTEGER N
      INTEGL=0.
      WRITE(*,*)' ENTER A,B,N'
      READ(*,*) A,B,N
      H=(B-A)/N
      CALL SHOWFX(A,N,H)
      CALL SUMMATION(A,N,H,SUMODD,SUMEVN)
*------------------------------------------------------------
*     EQUATION 21.18  P.599
*------------------------------------------------------------
      INTEGL=(B-A)*(F(A)+4.*SUMODD+2.*SUMEVN+F(B))/(3.*N)
*------------------------------------------------------------
      WRITE(*,50)
50    FORMAT(9X,'N',13X,'H',11X,'F(A)',9X,'F(B)',6X,'INTEGRAL',/)
      WRITE(*,100) N,H,F(A),F(B),INTEGL
100   FORMAT(5X,I5,5X,5F13.7)
      STOP
      END
*=================================================================
*------------------------------------------------------------
      FUNCTION F(X)
      REAL X
      F = 0.2+25.*X-200.*X*X+675.*X**3.-900.*X**4.+400.*X**5.
      END
*------------------------------------------------------------
```

```
      SUBROUTINE SUMMATION(A,N,H,SUMODD,SUMEVN)
      REAL A,H,SUMODD,SUMEVN
      SUMODD =0.
      SUMEVN =0.
      DO 10 I=1,N,2
        IF(I.GE.N) GO TO 30
        SUMODD=SUMODD+F(A+(I*H))
10    CONTINUE
30    DO 20 J=2,N,2
        IF(J.GE.N) GO TO 100
        SUMEVN=SUMEVN+F(A+(J*H))
20    CONTINUE
*------------------------------------------
100   WRITE(*,50)
50    FORMAT(//9X,'SUMODD',6X,'SUMEVEN',/)
      WRITE(*,200) SUMODD,SUMEVN
200   FORMAT(/5X,2F12.6//)
*------------------------------------------
      RETURN
      END
*============================================================
      SUBROUTINE SHOWFX(A,N,H)
      REAL A,H
       WRITE(*,50) F(A)
50     FORMAT(10X,'F(  0 ) =',F12.6)
       DO 10 I=1,N
        WRITE(*,100) I,F(A+(I*H))
10     CONTINUE
100    FORMAT(10X,'F(',I3,' )',' =',F12.6)
```

```
      RETURN
      END
*==============================================================
```

เอาท์พุทจากโปรแกรม คือ

```
  ENTER A,B,N
0
0.8
4
           F(  0 ) =     0.200000
           F(  1 ) =     1.288000
           F(  2 ) =     2.456000
           F(  3 ) =     3.464000
           F(  4 ) =     0.232004

         SUMODD      SUMEVEN

         4.752000    2.456000

         N              H            F(A)          F(B)        INTEGRAL

         4          0.2000000    0.2000000     0.2320040     1.6234670
```

จะได้อินทีเกรท = 1.6234670

ตัวอย่างที่ 9.5 จงหาอินทีเกรทในตัวอย่าง 9.1 โดยใช้วิธีการซิมป์สัน 1/3

วิธีทำ โปรแกรมที่ใช้ในการคำนวณจากตัวอย่าง 9.4 โดยเปลี่ยน f(x)

เอาท์พุทจากโปรแกรม คือ

a)

```
          PLEASE ENTER INTERVAL OF INTEGRATION A,B
0
3


          SINGLE SIMSON 1/3 RULE

          INTEGRATE F(X) = (1.-EXP(-X))

          FROM  0.00000 TO  3.00000

          F(  0 ) =    0.000000
          F(  1 ) =    0.776870
          F(  2 ) =    0.950213

          N             H            F(A)          F(B)        INTEGRAL

          2         1.5000000    0.0000000    0.9502130    2.0288463
```

b)

```
          PLEASE ENTER INTERVAL OF INTEGRATION A,B
-2
4


          SINGLE SIMSON 1/3 RULE

          INTEGRATE F(X) = 1.-X-4.*(X**3.)+X**5.

          FROM -2.00000 TO  4.00000

          F( X0 ) =    3.000000
          F( X1 ) =   -3.000000
          F( X2 ) =  765.000000

          N             H            F(A)          F(B)        INTEGRAL

          2         3.0000000    3.0000000  765.0000000  756.0000000
```

c)

```
SINGLE SIMSON 1/3 RULE

INTEGRATE F(X) = 8+4*SIN(X)

FROM  0.00000 TO  1.57080

F( X0 ) =    8.000000
F( X1 ) =   10.828427
F( X2 ) =   12.000000

N             H            F(A)          F(B)        INTEGRAL

2         0.7853982    8.0000000   12.0000000   16.5754910
```

ตัวอย่างที่ 9.6 จงหาค่าอินทีเกรทในตัวอย่าง 9.1 โดยวิธี Multiple simpson 1/3 rule โดยใช้ N = 6

วิธีทำ โปรแกรมที่ใช้ในการคำนวณ คือ

```
*       MULTIPLE SIMSON'S  1/3 RULES
*

        PROGRAM SIMPSN
        REAL A,B,INTEGL,SUMODD,SUMEVEN,H
        INTEGER N
        INTEGL=0.
        WRITE(*,20)
20       FORMAT(/,10X,'ENTER LEFT AND RIGHT ENDPOINTS ',
    +            'OF INTEGRATION A,B AND INTERVAL,N ')
        READ(*,*) A,B,N
        H=(B-A)/N
        CALL SHOWFX(A,N,H)
        CALL SUMMATION(A,N,H,SUMODD,SUMEVN)
*---------------------------------------------------------
*       EQUATION 21.18  P.599
*---------------------------------------------------------
        INTEGL=(B-A)*(F(A)+4.*SUMODD+2.*SUMEVN+F(B))/(3.*N)
*---------------------------------------------------------
        WRITE(*,10) A,B
        WRITE(*,50)
10       FORMAT(/, 10X,'THE MULTIPLE SIMSON 1/3 RULES',/,
    +             10X,'INTEGRATE   F(X) = (1.-EXP(-X))',/,
    +             10X,'FROM',F9.5,' TO',F9.5,/)
```

```
50       FORMAT(10X,'N',13X,'H',11X,'F(A)',9X,'F(B)',7X,
     +          'INTEGRAL',/)
         WRITE(*,100) N,H,F(A),F(B),INTEGL
100      FORMAT(6X,I5,6X,5F13.7)
         STOP
         END
*=================================================================
         SUBROUTINE SUMMATION(A,N,H,SUMODD,SUMEVN)
         REAL A,H,SUMODD,SUMEVN
         SUMODD =0.
         SUMEVN =0.
         DO 10 I=1,N,2
           IF(I.GE.N) GO TO 30
           SUMODD=SUMODD+F(A+(I*H))
10       CONTINUE
30       DO 20 J=2,N,2
           IF(J.GE.N) GO TO 100
           SUMEVN=SUMEVN+F(A+(J*H))
20       CONTINUE
*-------------------------------------------
100      WRITE(*,50)
50       FORMAT(/,10X,'SUM ODD',7X,'SUM EVEN')
         WRITE(*,200) SUMODD,SUMEVN
200      FORMAT(6X,2F12.6)
*-------------------------------------------
         RETURN
         END
*=================================================================
         SUBROUTINE SHOWFX(A,N,H)
```

```
      REAL A,H
        WRITE(*,50) F(A)
50        FORMAT(10X,'F( X0 ) =',F12.6)
        DO 10 I=1,N
          WRITE(*,100) I,F(A+(I*H))
10        CONTINUE
100       FORMAT(10X,'F( X',I2,' )',' =',F12.6)
      RETURN
      END
*===========================================================
*-----------------------------------------------------------
      FUNCTION F(X)
      REAL X
      F =  (1-EXP(-X))
      END
*-----------------------------------------------------------
```

เอาท์พุทจากโปรแกรม คือ

a)



```
          ENTER LEFT AND RIGHT ENDPOINTS OF INTEGRATION A,B AND INTERVAL,N
0
3
6
          F( X0 ) =     0.000000
          F( X 1 ) =     0.393469
          F( X 2 ) =     0.632121
          F( X 3 ) =     0.776870
          F( X 4 ) =     0.864665
          F( X 5 ) =     0.917915
          F( X 6 ) =     0.950213

          SUM ODD        SUM EVEN
          2.088254    1.496785

          THE MULTIPLE SIMSON 1/3 RULES
          INTEGRATE   F(X) = (1.-EXP(-X))
          FROM  0.00000 TO  3.00000

          N            H            F(A)          F(B)         INTEGRAL

          6         0.5000000    0.0000000     0.9502130     2.0494666
```

b)

```
          ENTER LEFT AND RIGHT ENDPOINTS OF INTEGRATION A,B AND INTERVAL,N
-2
4
6

          F( X0 ) =    3.000000
          F( X1 ) =    5.000000
          F( X2 ) =    1.000000
          F( X3 ) =   -3.000000
          F( X4 ) =   -1.000000
          F( X5 ) =  133.000000
          F( X6 ) =  765.000000

          SUM ODD       SUM EVEN
        135.000000    0.000000

          THE MULTIPLE SIMSON 1/3 RULES

          INTEGRATE F(X) = 1.-X-4.*(X**3.)+X**5.

          FROM -2.00000 TO  4.00000

          N             H           F(A)          F(B)        INTEGRAL

          6          1.0000000    3.0000000  765.0000000  436.0000000
```

c)

```
          PLEASE ENTER INTERVAL ,N
6
          F( X0 ) =    8.000000
          F( X1 ) =    9.035276
          F( X2 ) =   10.000000
          F( X3 ) =   10.828427
          F( X4 ) =   11.464102
          F( X5 ) =   11.863703
          F( X6 ) =   12.000000

          SUM ODD       SUM EVEN
          31.727407   21.464102

          THE MULTIPLE SIMSON 1/3 RULES

          INTEGRATE F(X) = (8.+4.*SIN(X))

          FROM  0.00000 TO  1.57080

          N             H           F(A)          F(B)        INTEGRAL

          6          0.2617994    8.0000000   12.0000000   16.5664768
```

9.3 การอินทีเกรทโดยแต่ละ segment มีค่าไม่เท่ากัน

$$I \cong h_1 \frac{f(x_0)+f(x_1)}{2} + h_2 \frac{f(x_1)+f(x_2)}{2} + \ldots.. + h_n \frac{f(x_{n-1})+f(x_n)}{2}$$

เมื่อ h_i ความกว้างของส่วนที่ i



รูปที่ 9.5 การอินทีเกรทโดยแต่ละส่วนมีค่าไม่เท่ากัน

9.4 การอินทีเกรทโดยวิธีรอมเบิร์ก

Romberg Integration ประยุกต์จาก trapezoidal rule

$$I = \frac{4}{3} I(h_2) - \frac{1}{3} I(h_1)$$

เมื่อ h_1, h_2 = step size

อัลกอริทึมที่ใช้กับคอมพิวเตอร์โดยวิธีรอมเบิร์ก คือ

$$I_{j,k} \cong \frac{4^{k-1} I_{j+1,k-1} - I_{j,k-1}}{4^{k-1} - 1}$$

เมื่อ $I_{j+1,k-1}$ และ $I_{j,k-1}$= อินทีเกรทที่ถูกต้องมากและน้อยตามลำดับ

ตัวอย่างที่ 9.7 จงอินทีเกรทสมการในตัวอย่าง 9.4 โดยใช้วิธี single และ multiple trapezoidal rule จากนั้นใช้วิธีของรอมเบิร์กเพื่อทำให้ผลเฉลยถูกต้องยิ่งขึ้น

วิธีทำ $I \cong \frac{4}{3}(1.4848) - \frac{1}{3}(1.0688)$

= 1.623467

ซึ่งมีความคลาดเคลื่อนเพียง 1%

โปรแกรมที่ใช้ในการคำนวณ โดยวิธีรอมเบิร์ก คือ

```
      PROGRAM ROMBERG
      DIMENSION R(10,10)
      REAL A,B
      WRITE(*,*)' PLEASE ENTER LIMIT OF INTEGRAL',
     +          ' A,B AND ERROR ES'
      READ(*,*) A,B,ES
      FX0=FUNC(A)
      FXN=FUNC(B)
      R(1,1)=(FX0+FXN)*(B-A)/2.
      DO 10 I=1,9
        N=2**I
        CALL TRAPEZ(A,B,N,AREA)
        R(I+1,1)=AREA
```

```
          DO 20 IC=2,I+1
            K =IC-1
            IR=2+I-IC
            R(IR,IC)=((4**K)*R(IR+1,K)-R(IR,K))/(4**K-1.)
20          CONTINUE
          ER=ABS(100.*(R(1,K+1)-R(2,K))/R(1,K+1))
          IF(ER.LE.ES) GO TO 30
10        CONTINUE
30        WRITE(*,100) A,B
100       FORMAT(//,7X,'INTEGRATE FUNCTION ',//,
     +    7X,'F(X)=0.2+25.*X-200.*X*X+675.*(X**3.)',
     +                 '-900.*(X**4.)+400.*(X**5.)',//
     +             7X,'FROM',F10.6,' TO',F10.6,/)
        WRITE(*,200) R(1,K+1),ER
200       FORMAT('  FINAL INTEGRAL VALUE  =',F16.10,/,
     +           '  WITH  RELATIVE ERROR  =',F16.10,' %')
        STOP
        END
*==============================================================
        SUBROUTINE TRAPEZ(A,B,N,AREA)
        H=(B-A)/N
        SUM=0.
        X=A+H
        DO 10 I=1,N-1
          FX=FUNC(X)
          SUM=SUM+FX
          X=X+H
10        CONTINUE
        FX0=FUNC(A)
```

```
      FXN=FUNC(B)
      AREA=(FX0+FXN+2.*SUM)*H/2.
      RETURN
      END
*============================================================
      FUNCTION FUNC(X)
      FUNC=0.2+25.*X-200.*X*X+675.*(X**3.)
     +     -900.*(X**4.)+400.*(X**5.)
      RETURN
      END
*============================================================
```

เอาท์พุท คือ

```
  PLEASE ENTER LIMIT OF INTEGRAL A,B AND ERROR ES
0
.8
.000001

       INTEGRATE FUNCTION

       F(X)=0.2+25.*X-200.*X*X+675.*(X**3.)-900.*(X**4.)+400.*(X**5.)

       FROM  0.000000 TO  0.800000

  FINAL INTEGRAL VALUE  =    1.6405341625
  WITH  RELATIVE ERROR  =    0.0000000000 %
```

จะได้ค่าอินทีเกรท = 1.6405341625

ค่าที่แท้จริง = 1.640533

มีค่าถูกต้องถึงทศนิยม 5 หลัก

แผนผังต่อไปนี้แสดงการคำนวณโดยวิธีรอมเบิร์ก โดย a) iteration ครั้งแรก b) iteration ครั้งที่สอง c) iteration ครั้งที่สาม

	$0(h^2)$	$0(h^4)$	$0(h^6)$	$0(h^8)$
a)	0.172800	1.367467		
	1.068800			
b)	0.172800	1.367467	1.640533	
	1.068800	1.623467		
	1.481800			
c)	0.172800	1.367467	1.640533	1.640533
	1.068800	1.623467	1.640533	
	1.481800	1.639467		
	1.600800			

ตัวอย่างที่ 9.8 จงอาศัยการอินทีเกรทโดยวิธีรอมเบิร์กหาค่า $\int_1^2 (x+1/x)^2 dx$

วิธีทำ ส่วนหนึ่งของโปรแกรมที่ใช้ในการคำนวณ คือ

```
*==========================================================
      FUNCTION FUNC(X)
       FUNC = (X+1./X)**2.
      RETURN
      END
*==========================================================
```

เอาท์พุทจากโปรแกรม คือ

```
  PLEASE ENTER LIMIT OF INTEGRAL A,B AND ERROR ES
1
2
.000001

      INTEGRATE FUNCTION

        F(X)=(X+1./X)**2.

      FROM  1.000000 TO  2.000000

  FINAL INTEGRAL VALUE  =     4.8333330154
  WITH  RELATIVE ERROR  =     0.0000000000 %
```

9.5 การอินทีเกรทโดยวิธีเกาส์ควอดราเจอร์ (Gauss Quadrature)

เป็นวิธีการเลือกจุดในการอินทีเกรทให้เหมาะสมเพื่อที่จะได้พื้นที่ใต้กราฟที่ถูกต้องมากยิ่งขึ้น



รูปที่ 9.6 วิธีเกาส์ควอดราเจอร์

ตัวอย่างที่ 9.9 จงอินทีเกรทสมการในตัวอย่าง 9.4 โดยวิธีเกาส์ควอดราเจอร์

วิธีทำ โปรแกรมที่ใช้ในการคำนวณ คือ

```
      PROGRAM GAUSS_QUADRATURE
      DIMENSION XI(21),W(21)
      DATA XI/0.0000000,-.5773503,.5773503,
     +        -.7745967,0.0000000,0.7745967,
     +        -.8611363,-.3399810,0.3399810,
     +        0.8611363,-.9061798,-.5384693,
     +        0.0000000,0.5384693,0.9061798,
     +        -.9324695,-.6612094,-.2386192,
     +        0.2386192,0.6612094,0.9324695/
      DATA W/2.0000000,1.0000000,1.0000000,
     +        0.5555556,0.8888889,0.5555556,
     +        0.3478549,0.6521452,0.6521452,
     +        0.3478549,0.2369269,0.4786287,
     +        0.5688889,0.4786287,0.2369269,
     +        0.1713245,0.3607616,0.4679139,
     +        0.4679139,0.3607616,0.1713245/
      REAL A,B
      WRITE(*,*)' PLEASE ENTER LIMIT OF INTEGRAL A,B '
      READ(*,*) A,B
      A0=(A+B)/2.
      A1=(B-A)/2.
      IC=1
      DO 10 NG=1,6
        SUM=0.
```

```
            DO 20 ITERMS=1,NG
              X=A0+A1*XI(IC)
              AI=FUNC(X)
              SUM=SUM+W(IC)*AI
              IC=IC+1
20           CONTINUE
            SUM=A1*SUM
            WRITE(*,100) NG,SUM
100          FORMAT(' RESULT OF INTEGRATION WITH',I2,
    +             ' GUASS POINT(S) IS',F16.10)
10         CONTINUE
          STOP
          END
*-----------------------------------------------
          FUNCTION FUNC(X)
          FUNC=0.2+25.*X-200.*(X*X)+675*(X**3.)
    +          -900.*(X**4.)+400.*(X**5.)
          RETURN
          END
*-----------------------------------------------
```

เอาท์พุทจากโปรแกรม คือ

```
  PLEASE ENTER LIMIT OF INTEGRAL A,B
0
.8
 RESULT OF INTEGRATION WITH 1 GUASS POINT(S) IS     1.9648002386
 RESULT OF INTEGRATION WITH 2 GUASS POINT(S) IS     1.8225802183
 RESULT OF INTEGRATION WITH 3 GUASS POINT(S) IS     1.6405344009
 RESULT OF INTEGRATION WITH 4 GUASS POINT(S) IS     1.6405299902
 RESULT OF INTEGRATION WITH 5 GUASS POINT(S) IS     1.6405314207
 RESULT OF INTEGRATION WITH 6 GUASS POINT(S) IS     1.6405357122
```

ถ้าใช้ 6 จุดจะได้ค่าอินทีเกรท = 1.64053 ซึ่งตรงกับค่าที่แท้จริง

แบบฝึกหัดบทที่ 9

1. จงอินทีเกรทฟังก์ชันต่อไปนี้ โดยวิธีทราเปซอยดอล

$$\int_{-3}^{5} (4x+5)^3 dx$$

2. จงอินทีเกรทฟังก์ชันต่อไปนี้ โดยวิธีทราเปซอยดอล และวิธีซิมป์สัน

$$\int_{0}^{3} xe^{2x} dx$$

3. จงอินทีเกรทฟังก์ชันในข้อ 2. โดยวิธี multiple Simpson's rule

4. จงอินทีเกรทข้อมูลต่อไปนี้ โดยวิธีทราเปซอยดอล

x	0	0.1	0.2	0.3	0.4	0.5
y	1	7	4	3	5	2

5. ทำข้อ 4. โดยวิธีซิมป์สัน

6. จงหาค่าโดยวิธีรอมเบิร์ก

$$\int_{0}^{3} xe^{2x} dx$$

7. จงหาค่าโดยวิธีเกาส์ควอดราเจอร์

$$\int_{0}^{1} x^{0.1}(1.2-x)\left(1-e^{20(x-1)}\right)dx$$